*------------- Upper-Body Strength and Political Egalitarianism: Twelve Conceptual Replications ------------------------------------*
*----------------------------------------------------- STUDY 1 ---------------------------------------------------------------------*

*** Background variables
* Respondent age
generate age = 2011-s_13

*Respondents self-reported to be younger than 18 or older than 40 years of age are deleted (because the sample is suposed to be a student sample)
drop if age <=18
drop if age >=40

* Respondent sex
recode s_3 (1=0) (2=1), generate(sex)

label define sexLB 0 "Female" 1 "Male"
label values sex sexLB


**** Independent variables 
**Self-perceived physical strength - coded to 0-1 scale reflecting min and max values within sex.
generate Phys_Strength = s_5/100

summ Phys_Strength if sex == 0
generate phys_strength = (Phys_Strength-(r(min)))/(r(max)-r(min)) if sex == 0

summ Phys_Strength if sex == 1
replace phys_strength = (Phys_Strength-(r(min)))/(r(max)-r(min)) if sex == 1

bys sex: summ phys_strength


**** Dependent variables
**Social Dominance Orientation - coded to 0-1 scale reflecting min and max values within sex.
recode s_8_1 s_8_2 s_8_3 s_8_4 s_8_5 s_8_6 (1=7) (2=6) (3=5) (4=4) (5=3) (6=2) (7=1)
alpha s_8_1 s_8_2 s_8_3 s_8_4 s_8_5 s_8_6
generate sdo = ((s_8_1+s_8_2+s_8_3+s_8_4+s_8_5+s_8_6)-6)/36

summ sdo if sex == 0
generate SDO = (sdo-(r(min)))/(r(max)-r(min)) if sex == 0

summ sdo if sex == 1
replace SDO = (sdo-(r(min)))/(r(max)-r(min)) if sex == 1



**Rightwing Authoritarianism
alpha s_8_7 s_8_8 s_8_9 s_8_10 s_8_11 s_8_12

*Recodes reversed coded items - coded to 0-1 scale reflecting min and max values within sex.
recode s_8_7 s_8_8 s_8_9 s_8_10 (1=7) (2=6) (3=5) (4=4) (5=3) (6=2) (7=1)
alpha s_8_7 s_8_8 s_8_9 s_8_10 s_8_11 s_8_12

pwcorr s_8_7 s_8_8 s_8_9 s_8_10 s_8_11 s_8_12, sig

generate rwa = ((s_8_7+s_8_8+s_8_9+s_8_10+s_8_11+s_8_12)-6)/36
summ rwa if sex == 0
generate RWA = (rwa-(r(min)))/(r(max)-r(min)) if sex == 0

summ rwa if sex == 1
replace RWA = (rwa-(r(min)))/(r(max)-r(min)) if sex == 1



*----------------------------------------------------ANALYSES------------------------------------------------------------*

**** Reported descriptives:
tab sex if phys_strength !=.
summ age if phys_strength !=.

*Formidability
bys sex: summ phys_strength

*SDO
alpha s_8_1 s_8_2 s_8_3 s_8_4 s_8_5 s_8_6 if phys_strength !=.
bys sex: summ SDO if phys_strength !=.

*RWA
alpha s_8_7 s_8_8 s_8_9 s_8_10 s_8_11 s_8_12 if phys_strength !=.
bys sex: summ RWA if phys_strength !=.


**** Analyses
bys sex: reg SDO c.phys_strength c.age

bys sex: reg RWA c.phys_strength c.age
